


Planeetan liike kiinteän Auringon ympäri

Planeetan liike Auringon ympäri voidaan yksinkertaisimmassa muodossa mallintaa pitämällä Aurinkoa kiinteänä ja ajattelemalla ainoaksi vaikuttavaksi voimaksi Auringon ja planeetan välinen gravitaatio. Todellisuudessa aivan näin ei ole: Planeetta ja Aurinko muodostavat kappaleparin, jossa molemmat vaikuttavat toisiinsa ja kumpikin liikkuu systeemin yhteisen massakeskipisteen suhteen. Liikkuvia kappaleita on myös useampia, Auringon lisäksi monia planeettoja, ja kaikki vaikuttavat toisiinsa. Yksinkertainen malli on suhteellisen hyvä approksimaatio, jos voidaan ajatella, että Auringon massa on merkittävästi planeetan massaa suurempi ja muut planeetat ovat niin etäällä, että niiden vaikutus voidaan jättää huomiotta.



Gravitaatiolain mukaan Auringon ja planeetan välinen vetovoima on

$$F = \frac{\gamma m_1 m_2}{r^2},$$

missä m_1 on Auringon massa ja m_2 planeetan massa, γ on gravitaatiovakio ja r on kappaleiden välinen etäisyys. Oletetaan, että Aurinko sijaitsee origossa ja planeetan paikkavektori on $\vec{r}$. Tällöin planeettaan kohdistuva voima suuntautuu origoon. Newtonin lain mukaiseksi vektorimuotoiseksi liikeyhtälöksi saadaan

$$m_2 \vec{r}'' = -\frac{\gamma m_1 m_2}{r^2} \frac{\vec{r}}{r},$$

Sievennettynä tämä on

$$\vec{r}'' = -\frac{\alpha}{r^3} \vec{r},$$

missä $\alpha = \gamma m_1$. Kyseessä on toisen kertaluvun vektorimuotoinen yhtälö.

Jos oletetaan, että planeetan rata sijaitsee tasossa (mikä ei ole itsestään selvää, mutta seuraa kyllä liikeyhtälöstä), voidaan tason koordinaatteja merkitä x ja y , jolloin paikkavektori on

$\vec{r}(t) = x(t)\vec{i} + y(t)\vec{j}$. Differentiaaliyhtälö hajoaa tällöin kahdeksi komponenttiyhtälöksi

$$x'' = -\frac{\alpha x}{(x^2 + y^2)^{3/2}},$$
$$y'' = -\frac{\alpha y}{(x^2 + y^2)^{3/2}}.$$

Merkitsemällä $x' = u$ ja $y' = v$ voidaan tämä palauttaa ensimmäistä kertalukua olevaksi neljän yhtälön normaaliryhmäksi, joka seuraavassa syötetään laskentaa varten *Mapleen*.

Aluksi kuitenkin on syytä hävittää mahdollisista aiemmista laskuista jääneet muuttujat:

```
> restart;
```

Normaaliryhmä on

```
> ryhmä:= diff(x(t), t)=u(t), diff(y(t), t)=v(t), diff(u(t),  
t)=-alpha*x(t)/(x(t)^2+y(t)^2)^(3/2), diff(v(t),  
t)=-alpha*y(t)/(x(t)^2 + y(t)^2)^(3/2);
```

```
ryhmä :=
```

$$\frac{d}{dt} x(t) = u(t), \frac{d}{dt} y(t) = v(t), \frac{d}{dt} u(t) = -\frac{\alpha x(t)}{(x(t)^2 + y(t)^2)^{(3/2)}}, \frac{d}{dt} v(t) = -\frac{\alpha y(t)}{(x(t)^2 + y(t)^2)^{(3/2)}}$$

ja probleeman tuntemattomat funktiot

```
> tuntemattomat:= x(t), y(t), u(t), v(t);
```

```
tuntemattomat := x(t), y(t), u(t), v(t)
```

Vakio α on Auringon massan ja gravitaatiovakion tulo (yksikkönä $\text{Nm}^2/\text{kg} = \text{m}^3/\text{s}^2$):

```
> alpha:= (1.989*10^30)*(6.673*10^(-11));
```

```
α := 0.1327259700 1021
```

Alkuehdot voidaan antaa vaikkapa jonkin planeetan mukaisesti. Maan keskietäisyys Auringosta on noin 150 miljoonaa kilometriä ja ratanopeus keskimäärin noin 30 kilometriä sekunnissa. Seuraavat alkuehdot ovat tämän mukaisia. Yksikköinä on käytettävä metrejä ja sekunteja kuten eo. vakiossakin:

```
> alkuehto:= x(0)=0, y(0)=150*10^9, u(0)=30*10^3, v(0)=0;  
      alkuehto := x(0) = 0, y(0) = 150000000000, u(0) = 30000, v(0) = 0
```

Luontevaa on laskea rata yhden Maan vuoden ajalta yksikkönä sekunti:

```
> tmax:= 365*24*60*60;  
      tmax := 31536000
```

Differentiaaliyhtälöryhmän ratkaisu numeerisesti:


```
> rtk:= dsolve({ryhma, alkuehto}, {tuntemattomat}, type=numeric,  
      output=listprocedure);  
rtk := [t = (proc(t) ... end proc), u(t) = (proc(t) ... end proc),  
      v(t) = (proc(t) ... end proc), x(t) = (proc(t) ... end proc), y(t) = (proc(t) ... end proc)]
```

Poimitaan ratkaisusta planeetan paikkaa ilmaisevat komponentit listaksi.

```
> rata:= subs(rtk, [x(t), y(t)]):
```

Planeetan rata voidaan piirtää parametrikäyränä; Aurinko sijaitsee origossa.

```
> plot([rata[1], rata[2], 0..tmax], scaling=constrained);
```




Ratanopeuden vaihtelusta voidaan myös piirtää kuvaaja:

```
> ratanopeus:= subs(rtk, sqrt(u(t)^2+v(t)^2));
```

$$ratanopeus := \sqrt{(\text{proc}(t) \dots \text{end proc})^2 + (\text{proc}(t) \dots \text{end proc})^2}$$

```
> plot(ratanopeus, 0..tmax, view=[0..tmax, 0..30000]);
```



>

Tehtävä

Tutki, miten alkuehdossa annettu nopeus vaikuttaa rataa. Miten käy, jos nopeutta a) pienennetään, b) suurennetaan, c) sen suuntaa muutetaan? Mitä tapahtuu, jos nopeuden molemmat komponentit ovat = 0? Taivaankappaleiden radat tässä yksinkertaisessa mallissa voivat olla ellipsejä (erikoistapauksena ympyröitä), paraabeleja ja hyperbelejä. Saatko alkuehtoja muuttamalla esiin eri ratatyypit?

Linkit

[normaaliryhmä](#)

JP & SKK & MS 12.07.2001